Processing で描ける図形まとめ

Processingで描ける図形についてまとめました。

この記事でよく使っている vertex について解説記事を書いたのでこちらもあわせてご覧ください！
[Processingのvertexについて解説！ : だらっと学習帳](http://blog.livedoor.jp/reona396/archives/55712534.html)

p5.js版も書きました！こちらもどうぞ。
[p5\.jsで描ける図形まとめ \- Qiita](https://qiita.com/reona396/items/95812b213910c007154f)

# 円
[ellipse()](https://processing.org/reference/ellipse_.html) で描けます。
横の大きさと縦の大きさをそれぞれ指定できるので、楕円も描けます。
正円(真円)はellipseで横の大きさと縦の大きさを同じにするか、Processing3.5で追加された [circle\(\)](https://processing.org/reference/circle_.html) を利用すると描けます。
![daen.png](https://qiita-image-store.s3.ap-northeast-1.amazonaws.com/0/67311/0ff11c96-90ca-1d71-78bc-6bdb545803a6.png)

```java
void setup() {
  size(400, 400);
}

void draw() {
  background(255);

  // 中心線
  strokeWeight(1);
  stroke(200);
  line(0, height/2, width, height/2);
  line(width/2, 0, width/2, height);

  noFill();
  strokeWeight(2);

  // 赤い楕円
  stroke(200, 0, 0);
  ellipse(width/2, height/2, 350, 150);

  // 青い楕円
  stroke(0, 0, 200);
  ellipse(300, 300, 100, 150);

  // 緑の正円
  stroke(0, 200, 0);
  circle(width/2, height/2, 250);
}
```

# 四角
[rect()](https://processing.org/reference/rect_.html) で描けます。
横の大きさと縦の大きさを指定できるので長方形が描けます。
正方形はrectで横の大きさと縦の大きさを同じにするか、Processing3.5で追加された [square\(\)](https://processing.org/reference/square_.html)
を利用すると描けます。

[rectMode()](https://processing.org/reference/rectMode_.html) を使うと引数として指定した座標位置がどこを指すのか変更できます。
なお、デフォルトでは左上が指定されるので、`rectMode(CORNER)`と同じ状態になります。
![rect.png](https://qiita-image-store.s3.amazonaws.com/0/67311/ec0df711-0588-19bb-fb94-5e9100fdd962.png)

```java
void setup() {
  size(400, 400);
}

void draw() {
  background(255);

  // 中心線
  strokeWeight(1);
  stroke(200);
  line(0, height/2, width, height/2);
  line(width/2, 0, width/2, height);

  noFill();
  strokeWeight(2);

  // 赤い長方形
  stroke(200, 0, 0);
  rectMode(CENTER);
  rect(width/2, height/2, 150, 300);

  //青い正方形
  stroke(0, 0, 200);
  square(width/2, height/2, 230);

  // 緑の長方形
  stroke(0, 200, 0);
  rectMode(CORNER);
  rect(width/2, height/2, 180, 100);
}
```

四角の角のまるみも指定できます。
位置やサイズの指定の後に4つの角のまるみを一括で指定したり、4つの角のまるみをそれぞれ指定したりできます。

![shikaku.png](https://qiita-image-store.s3.ap-northeast-1.amazonaws.com/0/67311/62b56d61-e511-7604-2193-b88cafdf0e4b.png)

```java
void setup() {
  size(400, 400);
}

void draw() {
  background(255);

  // 中心線
  strokeWeight(1);
  stroke(200);
  line(0, height/2, width, height/2);
  line(width/2, 0, width/2, height);

  noFill();
  strokeWeight(2);

  // 赤い長方形、角まる
  stroke(200, 0, 0);
  rectMode(CENTER);
  rect(width/2, height/2, 150, 300, 10);

  // 青い長方形、角いろいろ
  stroke(0, 0, 200);
  rect(width/2, height/2, 300, 200, 50, 0, 10, 0);
}
```

# 半円、パイカット図形
半円もしくはパイやピザをカットしたような図形は [arc()](https://processing.org/reference/arc_.html) を使って描けます。
オプションとしてOPEN(弧の線だけを描いて図形を閉じきらない)、CHORD(弧の線とその間の直線で図形を閉じきる)、PIE(弧と始点の間を線で結びピース状に閉じきる)が用意されています。

![pict.png](https://qiita-image-store.s3.amazonaws.com/0/67311/da14f3e4-f0e0-1e0a-8d3f-9829a523ea00.png)

```java
void setup() {
  size(400, 400);
}

void draw() {
  background(255);

  // 中心線
  strokeWeight(1);
  stroke(200);
  line(width/2, 0, width/2, height);
  line(0, height/2, width, height/2);

  strokeWeight(2);

  // 左上の赤い半円。線は閉じきっていない
  stroke(200, 0, 0);
  fill(200, 150, 150);
  arc(width/2, height/2, 250, 250, -PI, -PI/2, OPEN);

  // 下の青い半円。線は閉じきっている
  stroke(0, 0, 200);
  fill(150, 150, 200);
  arc(width/2, height/2, 250, 250, 0, PI, CHORD);

  // 右上の緑のピース。線は始点を含めて閉じきっている
  stroke(0, 200, 0);
  fill(150, 200, 150);
  arc(width/2, height/2, 250, 250, radians(270), radians(320), PIE);
}
```

# 三角形
[triangle()](https://processing.org/reference/triangle_.html) で描く方法もありますが、
[beginShape()](https://processing.org/reference/beginShape_.html)、[endShape()](https://processing.org/reference/endShape_.html)、[vertex()](https://processing.org/reference/vertex_.html) を使って描くと扱いやすいです。
詳しくは以下の記事をご覧ください。
[Processing で三角形を描く : だらっと学習帳](http://blog.livedoor.jp/reona396/archives/55107577.html)

![0127131617.png](https://qiita-image-store.s3.amazonaws.com/0/67311/1f284b23-afd5-fe7a-a32e-9b4c11f0f631.png)

```java
void setup() {
  size(400, 400);
}

void draw() {
  background(255);

  // 中心線
  strokeWeight(1);
  stroke(200);
  line(width/2, 0, width/2, height);
  line(0, height/2, width, height/2);

  // 灰色の円
  stroke(200);
  ellipse(width/2, height/2, 150*2, 150*2);

  // 三角形
  noFill();
  strokeWeight(2);
  stroke(200, 0, 0);
  drawTriangle(width/2, height/2, 150);  // 横の位置、縦の位置、円の半径
}

void drawTriangle(int x, int y, int r) {
  pushMatrix();
  translate(x, y);  // 中心となる座標
  rotate(radians(-90));

  // 円を均等に3分割する点を結び、三角形をつくる
  beginShape();
  for (int i = 0; i < 3; i++) {
    vertex(r*cos(radians(360*i/3)), r*sin(radians(360*i/3)));
  }
  endShape(CLOSE);

  popMatrix();
}
```

# ひし形
[beginShape()](https://processing.org/reference/beginShape_.html)、[endShape()](https://processing.org/reference/endShape_.html)、[vertex()](https://processing.org/reference/vertex_.html) を使って描けます。
詳しくは以下の記事をご覧ください。
[Processing で ひし形 を描く : だらっと学習帳](http://blog.livedoor.jp/reona396/archives/54824798.html)

![0127132923.png](https://qiita-image-store.s3.amazonaws.com/0/67311/554af988-f8df-e63c-6520-d8fa8360a418.png)

```java
void setup() {
  size(400, 400);
}

void draw() {
  background(255);

  // 中心線
  strokeWeight(1);
  stroke(200);
  line(width/2, 0, width/2, height);
  line(0, height/2, width, height/2);

  // ひし形
  noFill();
  strokeWeight(2);
  stroke(200, 0, 0);
  drawDiamond(width/2, height/2, 150);  // 横の位置、縦の位置、中心点と中心から遠い頂点までの距離
}

void drawDiamond(int x, int y, int r) {
  int R;
  pushMatrix();
  translate(x, y);
  beginShape();
  for (int i = 0; i < 4; i++) {
    if (i % 2 == 0) {
      R = r/2;
    } else {
      R = r;
    }
    vertex(R*cos(radians(90*i)), R*sin(radians(90*i)));
  }
  endShape(CLOSE);
  popMatrix();
}
```

# 正多角形
## シンプルな描き方
正五角形、正六角形、…正n角形は [beginShape()](https://processing.org/reference/beginShape_.html)、[endShape()](https://processing.org/reference/endShape_.html)、[vertex()](https://processing.org/reference/vertex_.html) を使って描けます。
詳しくは以下の記事をご覧ください。
[Processingで多角形・星を描く : だらっと学習帳](http://blog.livedoor.jp/reona396/archives/54602822.html)

正六角形を描く例 :

![0127134641.png](https://qiita-image-store.s3.amazonaws.com/0/67311/08f4749b-069a-ddbe-bf38-47f7dbdf8550.png)

```java
void setup() {
  size(400, 400);
}

void draw() {
  background(255);

  // 中心線
  strokeWeight(1);
  stroke(200);
  line(width/2, 0, width/2, height);
  line(0, height/2, width, height/2);

  // 正六角形
  noFill();
  strokeWeight(2);
  stroke(200, 0, 0);
  drawPolygon(width/2, height/2, 150, 6);  // 横の位置、縦の位置、中心点と頂点までの距離、頂点数
}

void drawPolygon(int x, int y, int r, int vertexNum) {
  pushMatrix();
  translate(x, y);
  beginShape();
  for (int i = 0; i < vertexNum; i++) {
    vertex(r*cos(radians(360*i/vertexNum)), r*sin(radians(360*i/vertexNum)));
  }
  endShape(CLOSE);
  popMatrix();
}
```

## 数式を利用した描き方
線が少しガタガタしてしまっていますが、数式を利用しても描けます。
数式は以下の記事を参考にさせていただきました。
[Parametric equation of a regular pentagon – GeoGebra](https://www.geogebra.org/m/cXXGKUQk)
[正多角形とスピログラフの数式 \| シキノート](http://slpr.sakura.ne.jp/qp/polygon-spirograph/)
詳しくは以下の記事をご覧ください。
[\[Processing\]\[p5\.js\]多角形や星をバシッとした数式で描く : だらっと学習帳](http://blog.livedoor.jp/reona396/archives/55768147.html)

![bashishiki01.png](https://qiita-image-store.s3.ap-northeast-1.amazonaws.com/0/67311/9169ac0c-ebed-c245-9fa7-698cd45d827e.png)

```java
void setup() {
  size(400, 400);
}

void draw() {
  background(255);

  // 中心線
  strokeWeight(1);
  stroke(200);
  line(width/2, 0, width/2, height);
  line(0, height/2, width, height/2);

  // 正六角形
  noFill();
  strokeWeight(2);
  stroke(200, 0, 0);
  drawPolygon(width/2, height/2, 150, 6);  // 横の位置、縦の位置、中心点と頂点までの距離、頂点数
}

void drawPolygon(int ox, int oy, int r, int vertexNum) {
  pushMatrix();
  translate(ox, oy);
  beginShape();
  for (int theta = 0; theta < 360; theta++) {
    PVector pos = calcPos(r, theta, vertexNum);

    float x = pos.x;
    float y = pos.y;

    vertex(x, y);
  }
  endShape(CLOSE);
  popMatrix();
}

PVector calcPos(int r, int t, int num) {
  float x = r * cos(radians(t)) * func(t, num);
  float y = r * sin(radians(t)) * func(t, num);

  return new PVector(x, y);
}

float func(int t, int num) {
  float A = cos(radians(180 / num));
  float b = 360 / num;
  float B = cos(radians(b * (t / b - floor(t / b)) - 180 / num));

  return A / B;
}
```

# 星
## シンプルな描き方
星は [beginShape()](https://processing.org/reference/beginShape_.html)、[endShape()](https://processing.org/reference/endShape_.html)、[vertex()](https://processing.org/reference/vertex_.html) を使って描けます。
トゲの数も調整できます。
詳しくは以下の記事をご覧ください。
[Processingで多角形・星を描く : だらっと学習帳](http://blog.livedoor.jp/reona396/archives/54602822.html)

トゲの数が5つの例 :
![0127140408.png](https://qiita-image-store.s3.amazonaws.com/0/67311/c57016e9-ee8e-c8ab-54b3-9a5566192d30.png)

```java
void setup() {
  size(400, 400);
}

void draw() {
  background(255);

  // 中心線
  strokeWeight(1);
  stroke(200);
  line(width/2, 0, width/2, height);
  line(0, height/2, width, height/2);

  // 星
  noFill();
  strokeWeight(2);
  stroke(200, 0, 0);
  drawStar(width/2, height/2, 150, 5);  // 横の位置、縦の位置、中心点とトゲの頂点までの距離、トゲの数
}

void drawStar(int x, int y, int r, int prickleNum) {
  int vertexNum = prickleNum*2;  // 頂点数(トゲの数*2)
  int R;  // 中心点から頂点までの距離
  int outR = r;  // 中心点からトゲまでの距離
  int inR = outR/2;  // 中心点から谷までの距離

  pushMatrix();
  translate(x, y);
  rotate(radians(-90));
  beginShape();
  for (int i = 0; i < vertexNum; i++) {
    if (i%2 == 0) {
      R = outR;
    } else {
      R = inR;
    }
    vertex(R*cos(radians(360*i/vertexNum)), R*sin(radians(360*i/vertexNum)));
  }
  endShape(CLOSE);
  popMatrix();
}
```

## 数式を利用した描き方
線が少しガタガタしてしまっていますが、数式を利用しても描けます。
 数式は以下の記事を参考にさせていただきました。
 [正多角形とスピログラフの数式 \| シキノート](http://slpr.sakura.ne.jp/qp/polygon-spirograph/)
 詳しくは以下の記事をご覧ください。
 [\[Processing\]\[p5\.js\]多角形や星をバシッとした数式で描く : だらっと学習帳](http://blog.livedoor.jp/reona396/archives/55768147.html)
 
![bashishiki02.png](https://qiita-image-store.s3.ap-northeast-1.amazonaws.com/0/67311/5cd6569a-1ffb-3781-a974-cf93c5a1c61f.png)

```java
void setup() {
  size(400, 400);
}

void draw() {
  background(255);

  // 中心線
  strokeWeight(1);
  stroke(200);
  line(width/2, 0, width/2, height);
  line(0, height/2, width, height/2);

  // 星
  noFill();
  strokeWeight(2);
  stroke(200, 0, 0);
  drawStar(width/2, height/2, 150, 5);  // 横の位置、縦の位置、中心点とトゲの頂点までの距離、トゲの数
}

void drawStar(int ox, int oy, int r, int prickleNum) {
  pushMatrix();
  translate(ox, oy);
  rotate(radians(-90));
  beginShape();
  for (int theta = 0; theta < 360; theta++) {
    PVector pos = calcPos(r, theta, prickleNum);

    float x = pos.x;
    float y = pos.y;

    vertex(x, y);
  }
  endShape(CLOSE);
  popMatrix();
}

PVector calcPos(int r, int t, int num) {
  float x = r * cos(radians(t)) * func(t, num);
  float y = r * sin(radians(t)) * func(t, num);

  return new PVector(x, y);
}

float func(int t, int num) {  
  float a = 360 / num;
  float A = cos(radians(a));
  float b = acos(cos(radians(num * t)));
  float B = cos(radians(a) - b / num);

  return A / B;
}
```

# キラキラ星(アステロイド曲線)
キラキラな星はアステロイド曲線と呼ばれ、[beginShape()](https://processing.org/reference/beginShape_.html)、[endShape()](https://processing.org/reference/endShape_.html)、[vertex()](https://processing.org/reference/vertex_.html) を使って描けます。
図形を縦方向に大きめに描くとよりそれらしくなります。
詳しくは以下の記事を以下の記事をご覧ください。
[Processing でキラキラ図形を描く : だらっと学習帳](http://blog.livedoor.jp/reona396/archives/55010192.html)

![0129151559.png](https://qiita-image-store.s3.amazonaws.com/0/67311/7d9025af-b879-f5c3-818e-4dabccfa2d2a.png)

```java
void setup() {
  size(400, 400);
}

void draw() {
  background(255);

  // 中心線
  strokeWeight(1);
  stroke(200);
  line(width/2, 0, width/2, height);
  line(0, height/2, width, height/2);

  // アステロイド曲線
  noFill();
  strokeWeight(2);
  stroke(200, 0, 0);
  drawTwinkleStar(width/2, height/2, 120);  // 横の位置、縦の位置、中心点と横方向の頂点までの距離
}

void drawTwinkleStar(int x, int y, int r) {
  pushMatrix();
  translate(x, y);
  beginShape();
  for (int theta = 0; theta < 360; theta++) {
    vertex(r * pow(cos(radians(theta)), 3), r * 1.4 * pow(sin(radians(theta)), 3));
  }
  endShape(CLOSE);
  popMatrix();
}
```

# びっくり吹き出し(アステロイド曲線2)
漫画でよく見かける
＿人人人人人人人人人人＿
＞　びっくり吹き出し　＜
￣Y^Y^Y^Y^Y^Y^Y^Y^Y￣
はアステロイド曲線で表現でき、[beginShape()](https://processing.org/reference/beginShape_.html)、[endShape()](https://processing.org/reference/endShape_.html)、[vertex()](https://processing.org/reference/vertex_.html) を使って描けます。
詳しくは以下の記事をご覧ください。
[Processing でびっくり吹き出しみたいなとげの多い図形(アステロイド曲線)を描く : だらっと学習帳](http://blog.livedoor.jp/reona396/archives/55522114.html)

![2016622132416.png](https://qiita-image-store.s3.amazonaws.com/0/67311/4d4c5d36-27f1-176f-f047-947b525eaa10.png)

```java
void setup() {
  size(400, 400);
}

void draw() {
  background(255);

  // 中心線
  strokeWeight(1);
  stroke(200);
  line(width/2, 0, width/2, height);
  line(0, height/2, width, height/2);

  // アステロイド曲線
  noFill();
  strokeWeight(2);
  stroke(200, 0, 0);
  strokeJoin(ROUND);
  drawAstroid(width/2, height/2, 20, 8);  // 横の位置、縦の位置、大きさ調整用変数、トゲの数
}

void drawAstroid(int ox, int oy, int r, int vertexNum) {
  vertexNum -= 1;
  pushMatrix();
  translate(ox, oy);
  beginShape();
  for (int theta = 0; theta < 360; theta++) {
    float x = r * (vertexNum * cos(radians(theta)) + cos(radians(-vertexNum * theta)));
    float y = r * (vertexNum * sin(radians(theta)) + sin(radians(-vertexNum * theta)));

    vertex(x, y);
  }
  endShape();
  popMatrix();
}
```

# ハート
ハートは [beginShape()](https://processing.org/reference/beginShape_.html)、[endShape()](https://processing.org/reference/endShape_.html)、[vertex()](https://processing.org/reference/vertex_.html) を使って描けます。
[strokeJoin()](https://www.processing.org/reference/strokeJoin_.html)を利用して線のつなぎ目をなめらかにしています。
詳しくは以下の記事をご覧ください。
[Processing でハートを描く : だらっと学習帳](http://blog.livedoor.jp/reona396/archives/55116805.html)

![0129152304.png](https://qiita-image-store.s3.amazonaws.com/0/67311/9551caec-c5e2-b54b-3ab0-8ab55a6a9e20.png)

```java
void setup() {
  size(400, 400);
}

void draw() {
  background(255);

  // 中心線
  strokeWeight(1);
  stroke(200);
  line(width/2, 0, width/2, height);
  line(0, height/2, width, height/2);

  // ハート
  noFill();
  strokeWeight(2);
  stroke(200, 0, 0);
  strokeJoin(ROUND); //線のつなぎ目について設定
  drawHeart(width/2, height/2, 8);  // 横の位置、縦の位置、大きさ調整用変数
}

void drawHeart(int ox, int oy, int size) {
  pushMatrix();
  translate(ox, oy);
  beginShape();
  for (int theta = 0; theta < 360; theta++) {
    float x = size * (16 * sin(radians(theta)) * sin(radians(theta)) * sin(radians(theta)));
    float y = (-1) * size * (13 * cos(radians(theta)) - 5 * cos(radians(2 * theta))
      - 2 * cos(radians(3 * theta)) - cos(radians(4 * theta)));

    vertex(x, y);
  }
  endShape(CLOSE);
  popMatrix();
}
```

# 花
お花は [beginShape()](https://processing.org/reference/beginShape_.html)、[endShape()](https://processing.org/reference/endShape_.html)、[curveVertex()](https://www.processing.org/reference/curveVertex_.html) を使って描けます。
詳しくは以下の記事をご覧ください。
[Processing でスーパーバラ曲線を描いてみた : だらっと学習帳](http://blog.livedoor.jp/reona396/archives/54660457.html)
![flower.png](https://qiita-image-store.s3.amazonaws.com/0/67311/eafab002-22e9-0469-b268-1b8b19cb054c.png)

```java
void setup() {
  size(400, 400);
}

void draw() {
  background(255);

  // 中心線
  strokeWeight(1);
  stroke(200);
  line(width/2, 0, width/2, height);
  line(0, height/2, width, height/2);

  // 花
  noFill();
  strokeWeight(2);
  stroke(200, 0, 0);
  drawFlower(width/2, height/2, 100);  // 横の位置、縦の位置、大きさ調整用変数
}

void drawFlower(int ox, int oy, int r) {
  pushMatrix();
  translate(ox, oy);
  beginShape();
  for (int theta = 0; theta < 360; theta++) {
    float R = r * abs(sin(radians(theta*5))) + r/2;
    float x = R * cos(radians(theta));
    float y = R * sin(radians(theta));

    curveVertex(x, y);
  }
  endShape(CLOSE);
  popMatrix();
}
```

# 桜
## 花びらが分かれているタイプ
花びらがそれぞれ分かれていて、先端に鋭い切れ込みの入った桜の花は、[beginShape()](https://processing.org/reference/beginShape_.html)、[endShape()](https://processing.org/reference/endShape_.html)、[vertex()](https://processing.org/reference/vertex_.html) を使って描けます。
数式は以下の記事を参考にさせていただきました。
[花の曲線を描く \- CinderellaJapan](https://sites.google.com/site/cinderellajapan/huanocg/huano-qu-xian)

![pict.png](https://qiita-image-store.s3.amazonaws.com/0/67311/0dd3c3a7-8286-51ec-32fa-aa9f370d3259.png)

```java
void setup() {
  size(400, 400);
}

void draw() {
  background(255);

  // 中心線
  strokeWeight(1);
  stroke(200);
  line(width/2, 0, width/2, height);
  line(0, height/2, width, height/2);

  // 桜
  noFill();
  strokeWeight(2);
  stroke(200, 0, 0);
  drawSakura(width/2, height/2, 200);  // 横の位置、縦の位置、大きさ調整用変数
}

void drawSakura(int ox, int oy, int or) {
  int petalNum = 5;  // 花びらの数

  pushMatrix();
  translate(ox, oy);
  rotate(radians(90));
  beginShape();
  for (int theta = 0; theta < 360; theta++) {
    float A = petalNum / PI * radians(theta);
    int md = floor(A) % 2;
    float r = pow(-1, md) * (A - floor(A)) + md;
    float R = r + 2 * calcH(r);

    float x = or * R * cos(radians(theta));
    float y = or * R * sin(radians(theta));

    vertex(x, y);
  }
  endShape(CLOSE);
  popMatrix();
}

float calcH(float x) {
  if (x < 0.8) {
    return 0;
  } else {
    return 0.8 - x;
  }
}
```

## 花びらが分かれていないタイプ
花びらが分かれていない、丸みのある桜の花は、[beginShape()](https://processing.org/reference/beginShape_.html)、[endShape()](https://processing.org/reference/endShape_.html)、[vertex()](https://processing.org/reference/vertex_.html) を使って描けます。
数式は以下の記事を参考にさせていただきました。
[～スーパーバラ曲線を作ろう～](https://img.atwikiimg.com/www42.atwiki.jp/syugyou/attach/250/1057/bar.html)

![image.png](https://qiita-image-store.s3.amazonaws.com/0/67311/235c9bd4-afb6-b96a-1fd7-68952567343c.png)

```java
void setup() {
  size(400, 400);
}

void draw() {
  background(255);

  // 中心線
  strokeWeight(1);
  stroke(200);
  line(width/2, 0, width/2, height);
  line(0, height/2, width, height/2);

  // 桜
  noFill();
  strokeWeight(2);
  stroke(200, 0, 0);
  drawSakura(width/2, height/2, 130);  // 横の位置、縦の位置、大きさ調整用変数
}

void drawSakura(int ox, int oy, int or) {
  pushMatrix();
  translate(ox, oy);
  beginShape();
  for (int theta = 0; theta < 360; theta++) {
    float A = (sin(radians(theta * 5)) + cos(radians(theta * 10))) / 2.0;
    float B = A * 0.5 + 1.0;
    float R = or * B;
    float x = R * sin(radians(theta + 90));
    float y = R * cos(radians(theta + 90));

    vertex(x, y);
  }
  endShape(CLOSE);
  popMatrix();
}
```

# 葉 (麻の葉、紅葉、楓)
葉っぱは [beginShape()](https://processing.org/reference/beginShape_.html)、[endShape()](https://processing.org/reference/endShape_.html)、[vertex()](https://www.processing.org/reference/vertex_.html) を使って描けます。
詳しくは以下の記事をご覧ください。
[Processing で 葉っぱを描く(麻の葉？紅葉、楓？) : だらっと学習帳](http://blog.livedoor.jp/reona396/archives/55372667.html)

![0922165413.png](https://qiita-image-store.s3.amazonaws.com/0/67311/1f82abd7-9e45-30ec-a523-b9263ab16c96.png)


```java
void setup() {
  size(400, 400);
}

void draw() {
  background(255);

  // 中心線
  strokeWeight(1);
  stroke(200);
  line(width/2, 0, width/2, height);
  line(0, height/2, width, height/2);

  // 葉っぱ
  noFill();
  strokeWeight(2);
  stroke(200, 0, 0);
  noFill();
  drawLeaf(width/2, height/2, 45);  // 横の位置、縦の位置、大きさ調整用変数
}

void drawLeaf(int ox, int oy, int r) {
  pushMatrix();
  translate(ox, oy);
  beginShape();
  for (int theta = 0; theta < 360; theta++) {
    float R = - r * (1 + (9.0/10.0) * cos(radians(8 * theta))) * (1 + (1.0/10.0) * cos(radians(24 * theta))) * ((9.0/10.0) + (1.0/10.0) * cos(radians(200 * theta))) * (1 + sin(radians(theta)));

    float x = R * cos(radians(theta));
    float y = R * sin(radians(theta));

    vertex(x, y);
  }
  endShape(CLOSE);
  popMatrix();
}
```

# しずく
涙のようなしずく型の図形は [beginShape()](https://processing.org/reference/beginShape_.html)、[endShape()](https://processing.org/reference/endShape_.html)、[vertex()](https://processing.org/reference/vertex_.html) を使って描けます。
コード中、`A`という変数がありますが、この値を小さくすると丸みのある形になり、大きくすると長細く鋭い形になります。
詳しくは以下の記事をご覧ください。
[Processing でしずく型曲線を描く : だらっと学習帳](http://blog.livedoor.jp/reona396/archives/54891368.html)

![0129154223.png](https://qiita-image-store.s3.amazonaws.com/0/67311/e38ad28b-bdca-b883-12d2-37e18376f4d7.png)

```java
void setup() {
  size(400, 400);
}

void draw() {
  background(255);

  // 中心線
  strokeWeight(1);
  stroke(200);
  line(width/2, 0, width/2, height);
  line(0, height/2, width, height/2);

  // しずく
  noFill();
  strokeWeight(2);
  stroke(200, 0, 0);
  drawDrop(width/2, height/2, 180, 5);  // 横の位置、縦の位置、大きさ調整用変数、鋭さ調整用変数
}

void drawDrop(int x, int y, int r, int A) {
  pushMatrix();
  translate(x, y);
  rotate(radians(-90));
  beginShape();
  for (int theta = 0; theta < 360; theta++) {
    float R = r / (A * sin(radians(theta)/2)+1);
    vertex(R * cos(radians(theta)), R * sin(radians(theta)));
  }
  endShape(CLOSE);
  popMatrix();
}
```

# 魚
魚の形は [beginShape()](https://processing.org/reference/beginShape_.html)、[endShape()](https://processing.org/reference/endShape_.html)、[vertex()](https://processing.org/reference/vertex_.html) を使って描けます。
数式は以下のページを参考にしました。
[Fish curve \- Wikipedia](https://en.wikipedia.org/wiki/Fish_curve)

![img.png](https://qiita-image-store.s3.ap-northeast-1.amazonaws.com/0/67311/e68c7e83-2403-1aa2-a783-dc77557800f4.png)

```java
void setup() {
  size(400, 400);
}

void draw() {
  background(255);

  // 中心線
  strokeWeight(1);
  stroke(200);
  line(width/2, 0, width/2, height);
  line(0, height/2, width, height/2);

  // 魚
  noFill();
  strokeWeight(2);
  stroke(200, 0, 0);
  drawFish(width/2, height/2, 150);  // 横の位置、縦の位置、大きさ調整用変数
}

void drawFish(int ox, int oy, int r) {
  pushMatrix();
  translate(ox, oy);
  beginShape();
  for (int theta= 0; theta< 360; theta++) {
    float x = r * cos(radians(theta)) - r * pow(sin(radians(theta)), 2) / sqrt(2);
    float y = r * cos(radians(theta)) * sin(radians(theta));

    vertex(x, y);
  }
  endShape(CLOSE);
  popMatrix();
}

```

# 穴あき図形
ドーナツのように穴のあいた図形は [beginContour()](https://processing.org/reference/beginContour_.html)、[beginShape()](https://processing.org/reference/beginShape_.html)、[endShape()](https://processing.org/reference/endShape_.html)、[vertex()](https://www.processing.org/reference/vertex_.html)、[endContour()](https://processing.org/reference/endContour_.html) を使って描けます。
詳しくは以下の記事をご覧ください。
[Processing / p5.js で穴あき図形を描く beginContour() : だらっと学習帳](http://blog.livedoor.jp/reona396/archives/55508707.html)

![pict.png](https://qiita-image-store.s3.amazonaws.com/0/67311/f35271b0-5485-3266-0ee7-c3830e03d2d5.png)

```java
float R; // 星(穴)の大きさ調整用
int vertexNum = 10; // 星の頂点数

int w = 300; //図形の外枠の横の長さ
int h = 300; //図形の外枠の縦の長さ

void setup() {
  size(400, 400);
}

void draw() {
  background(255);

  // 中心線
  strokeWeight(1);
  stroke(200);
  line(0, height/2, width, height/2);
  line(width/2, 0, width/2, height);

  // 星形穴あき図形
  noFill();
  strokeWeight(2);
  stroke(200, 0, 0);
  fill(200, 100, 100);

  pushMatrix();
  translate(width/2, height/2);
  rotate(radians(-90));
  beginShape();
  vertex(-w/2, -h/2);
  vertex(w/2, -h/2);
  vertex(w/2, h/2);
  vertex(-w/2, h/2);
  beginContour();
  for (int i = vertexNum; i > 0; i--) {
    if (i % 2 == 0) {
      R = w * 0.4;
    } else {
      R = w * 0.2;
    }
    vertex(R * cos(radians(360 * i / vertexNum)), R * sin(radians(360 * i / vertexNum)));
  }
  endContour();
  endShape(CLOSE);
  popMatrix();
}
```